

Assignment 2, Question 2 Multivariate Polynomials

Part (a)

```
> A := 6*y^2*x^3+2*x^2*y^2+5*y*x^2+3*x*y^2+y*x+y^2+x+y;
```

$$A := 6x^3y^2 + 2x^2y^2 + 5x^2y + 3xy^2 + xy + y^2 + x + y$$

```
> B := 2*y*x^2+x+y;
```

$$B := 2x^2y + x + y$$

We will do the division in $\mathbb{Z}[y][x]$ recursively. To write A in $\mathbb{Z}[y][x]$

```
> A := collect(A,x);
```

$$A := 6y^2x^3 + (2y^2 + 5y)x^2 + (3y^2 + y + 1)x + y^2 + y$$

So we divide $\text{lc}(A)$ by $\text{lc}(B)$ recursively in $\mathbb{Z}[y]$ and get $3y$.

```
> lcoeff(A,x);  
lcoeff(B,x);
```

$$6y^2$$
$$2y$$

We divide $\text{lm}(A)=x^3$ by $\text{lm}(B)=x^2$ to get the first term of the quotient is

```
> term := 3*y*x;  
Q := term;
```

$$\text{term} := 3yx$$
$$Q := 3yx$$

```
> A := expand(A-term*B);
```

$$A := 2x^2y^2 + 2x^2y + xy + y^2 + x + y$$

```
> A := collect(A,x);
```

$$A := (2y^2 + 2y)x^2 + (y + 1)x + y^2 + y$$

We repeat this. We divide $\text{lc}(A) = 2 \cdot y^2 + 2 \cdot y$ by $\text{lc}(B) = 2 \cdot y$

```
> divide(lcoeff(A,x),lcoeff(B,x),'term'); term;  
true
```

$$y + 1$$

```
> A := expand(A-term*B);
```

A := 0

So B | A and the quotient Q =

> Q := Q+term;

Q := 3xy + y + 1

Part (b)

My code for DIVIDE below is based on modifying the univariate division code

```
> QUOREM := proc(A,B,x) # division in F[x] e.g. Q[x]
  local R,Q,L,T;
  R := expand(A);
  Q := 0;
  while R <> 0 and degree(R,x) >= degree(B,x) do
    L := lcoeff(R,x)/lcoeff(B,x);
    T := L*x^(degree(R,x)-degree(B,x));
    Q := Q+T;
    R := expand( R-T*B );
  od;
  return Q,R;
end:

> DIVIDE := proc(A,B) # division in Z[x1,x2,...,xn]
  local X,x,R,Q,L,T;
  X := indets(A) union indets(B);
  if X={} then # base of recursion: A,B are integers
    if irem(A,B)=0 then return iquo(A,B); else return FAIL; fi;
  fi;
  x := X[1];
  R := expand(A);
  Q := 0;
  while R <> 0 and degree(R,x) >= degree(B,x) do
    L := DIVIDE(lcoeff(R,x),lcoeff(B,x));
    if T=FAIL then return FAIL fi;
    T := L*x^(degree(R,x)-degree(B,x));
    Q := Q+T;
    R := expand( R-T*B );
  od;
  if R=0 then return Q; else return FAIL; fi;
end:

> A := (6*y^2-5*y*z+z^2)*x^2+(7*y^2*z-3*y*z^2)*x+2*y^2*z^2;
  B := (2*y-z)*x+y*z;
      A := (6y2 - 5yz + z2) x2 + (7y2z - 3yz2) x + 2y2z2
      B := (2y - z) x + yz
```

```
> Q := DIVIDE(A,B);
```

$Q := (3y - z)x + 2yz$

```
> expand( A-B*Q );
```

0

```
> Q := DIVIDE(A+x,B);
```

$Q := \text{FAIL}$

```
> Q := DIVIDE(A+2,B);
```

$Q := \text{FAIL}$